



IVI Instrument Driver Programming Guide (Delphi Edition)

June 2012 Revision 2.0

1- Overview

1-1 Recommendation Of IVI-COM Driver

Although it is hard to say that Embarcadero Delphi is suitable for using IVI-COM instrument drivers, it is still possible to use COM components with COM wrapper classes. Comparing with Microsoft development environment, there are a lot of weak points for using Microsoft COM, however IVI-COM is still easier to use than using IVI-C instrument drivers that do not provide Pascal header files.

Therefore this guidebook recommends using IVI-COM instrument drivers through the COM wrapper classes.

Notes:

- This guidebook shows examples that use KikusuiPwx IVI instrument driver (KIKUSUI Pwx series DC Power Supply). You can also use IVI drivers for other vendors and other models in the same manner.
- This guidebook describes how to create 32bit (x86) programs that run under Windows7 (x64), using Embarcadero Delphi XE2.

1-2 IVI Instrument Class Interface

When using an IVI instrument driver, there are two approaches – using specific interfaces and using class interfaces. The former is to use interfaces that are specific to an instrument driver and you can utilize the most of features of the instrument. The later is to utilize instrument class interfaces that are defined in the IVI specifications allowing to utilize interchangeability features, but instrument specific features are restricted.

Notes:

- The instrument class to which the instrument driver belongs is documented in Readme.txt for each of drivers. The Readme document can be viewed from Start button → All Programs → Kikusui → KikusuiPwx menu.
- If the instrument driver does not belong to any instrument classes, you can't utilize class interfaces. This means that you cannot develop applications that utilize interchangeability features.

2- Example Using Specific Interface

Here we introduce an example using specific interfaces. By using specific interfaces, you can utilize the maximum feature (or model specific functions) provided by the driver but you have to spoil interchangeability.

2-1 Creating Application Project


To simplify explanation, this guidebook shows you an example of the simplest console application. After launching Delphi XE2 IDE, choose **File | New | Other...** menu to bring up the **New Items** dialogue. Select **Delphi Projects** and then **Console Application**.

The project has no file names immediately after it is created. Choose **File | Save Project As...** menu then give **guideAppDelphi** as the project file name. As for Delphi, the main program is placed in the .dpr file having the same file name with project.

2-2 Importing Type Libraries

What you should do first after creating a new project is, import type libraries for the IVI-COM instrument driver that you want to use. To import them, use a command-line tool TLIBIMP.EXE (bundled with Delphi).

Launch Command Prompt, and then use cd (change directory) command to go to the directory where the project files previously created are located. If the current directory display (using dir command) shows the project previously saved, it is the correct location.



```
cs: Command Prompt
E:\_JunkWork\guide\delphi\1>dir
Volume in drive E is kamakura
Volume Serial Number is AA00-0550

Directory of E:\_JunkWork\guide\delphi\1

2012/06/13  16:17    <DIR>          .
2012/06/13  16:17    <DIR>          ..
2012/06/13  16:17                254 guideAppDelphi.dpr
2012/06/13  16:17                7,037 guideAppDelphi.dproj
2012/06/13  16:17                285 guideAppDelphi.dproj.local
                3 File(s)      7,576 bytes
                2 Dir(s)  82,977,234,944 bytes free


E:\_JunkWork\guide\delphi\1>_
```

Figure 2-1 Working At Project Directory

Now execute the following command on the Command Prompt.

```
tlibimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/kipwx.dll"
```

Then, multiple wrapper modules related to IVI and VISA will be generated. (Although a lot of files are generated, you don't use all files.)



```

C:\> Command Prompt

E:\_JunkWork\guide\delphi\1>tlibimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/kipwx.dll"
Embarcadero TLBIMP Version 12.16581
Copyright(c) 1995-2010 Embarcadero Technologies, Inc.

Opening C:/Program Files (x86)/IVI Foundation/IVI/Bin/kipwx.dll
Type library loaded ...
Created E:\_JunkWork\guide\delphi\1\IviDriverLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\1\IviDriverLib_TLB.pas
Created E:\_JunkWork\guide\delphi\1\VisaComLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\1\VisaComLib_TLB.pas
Created E:\_JunkWork\guide\delphi\1\IviDCPwrLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\1\IviDCPwrLib_TLB.pas
Created E:\_JunkWork\guide\delphi\1\KikusuiPwxLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\1\KikusuiPwxLib_TLB.pas

E:\_JunkWork\guide\delphi\1>_

```

Figure 2-2 Generating kipwx TypeLib wrapper

Again back to the Delphi IDE, add the files to your project.

Right-click on **guideAppDelphi.exe** name in the Project Manager window to show the context menu, select **Add...** menu, select **KikusuiPwxLib_TLB.pas** from the previously generated files, then add it to the project. It will be okay if the file has been added in the Project Manager.

The type library wrapper module to add contains dependency to VCL (Visual Component Library). Therefore, when adding such module to the project that does not contain dependency to VCL, a confirmation message like below may be shown. Dependency to VCL is required so select Yes.




Figure 2-3 Confirmation for Adding VCL




Figure 2-4 Adding TLB Wrapper

It will be okay if the file has been added in the Project Manager. Also, confirm that **KikusuiPwxLib_TLB** is automatically added at **uses** clause.

```
uses
  System.SysUtils,
  KikusuiPwxLib_TLB in 'KikusuiPwxLib_TLB.pas';
```

Open the main program in the project, add **ComObj** and **windows** units to the existing **uses** clause.

```
uses
  System.SysUtils, ComObj, windows,
  KikusuiPwxLib_TLB in 'KikusuiPwxLib_TLB.pas';
```

2-3 Creating Object and Initializing Session

Add the following variable declarations at the **var** block in the main program.

```
var
  hr: HRESULT;
  instr: IKikusuiPwx;
```

In the main program's **try** block, write the following code fragments that opens a session for instrument driver object and close it. Here assume that an instrument (Kikusui Pwx series DC supply) having IP address 192.168.1.5 connected with LAN interface.

```
instr:= CoKikusuiPwx.Create;
hr:= instr.Initialize('TCPIP::192.168.1.5::INSTR', true, true, '');
hr:= instr.Close();
```

When creating a driver object, use **Create** method of **CoKikusuiPwx** component class. Just creating the object does not communicate with instrument, so furthermore invoke **Initialize** method and **Close** method.

Once the object is created with **Create**, its reference count is 1 at the point of time. When the scope of the variable **instr** has been lost (at the end of **try** block), the reference count is decremented. The object will be destroyed when the reference count becomes 0.

Now let's talk about the parameters for the **Initialize** method. Every IVI-COM instrument driver has an **Initialize** method that is defined in the IVI specifications. This method has the following parameters.

Table 2-1 Params of Initialize Method

Parameter	Type	Description
ResourceName	wideString	VISA resource name string. This is decided according to the I/O interface and/or address through which the instrument is connected. For example, a LAN-based instrument having IP address 192.168.1.5 will be TCP/IP::192.168.1.5::INSTR (when VXI-11 case).
IdQuery	wordBool	Specifying TRUE performs ID query to the instrument.
Reset	wordBool	Specifying TRUE resets the instrument settings.
OptionString	wideString	Overrides the following settings instead of default: RangeCheck Cache Simulate QueryInstrStatus RecordCoercions Interchange Check Furthermore you can specify driver-specific options if the driver supports DriverSetup features.

ResourceName specifies a VISA resource. If **IdQuery** is TRUE, the driver queries the instrument identities using a query command such as "***IDN?**". If **Reset** is TRUE, the driver resets the instrument settings using a reset command such as "***RST**".

OptionString has two features. One is what configures IVI-defined behaviours such as **RangeCheck**, **Cache**, **Simulate**, **QueryInstrStatus**, **RecordCoercions**, and **Interchange Check**. Another one is what specifies **DriverSetup** that may be differently defined by each of instrument drivers. Because the **OptionString** is a string parameter, these settings must be written as like the following example:

```
QueryInstrStatus = TRUE , Cache = TRUE , DriverSetup=12345
(DriverSetup=12345 is only an imaginary parameter for explanation.)
```

Names and setting values for the features being set are case-insensitive. Since the setting values are Boolean type, you can use any of TRUE, FALSE, 1, and 0. Use commas for splitting multiple items. If an item is not explicitly specified in the **OptionString** parameter, the IVI-defined default value is applied for the item. The IVI-defined default values are TRUE for **RangeCheck** and **Cache**, and FALSE for others.

Some instrument drivers may have special meanings for the **DriverSetup** parameter. It can specify items that are not defined by the IVI specifications when invoking the **Initialize** method, and its purpose and syntax are driver-specific. Therefore, specifying the **DriverSetup** must be at the last part on the **OptionString** parameter. Because the contents of **DriverSetup** are different depending on each driver, refer to driver's Readme document or online help.

2-4 Closing Session

To close instrument driver session, use the **Close** method.

2-5 Execution

You can execute the previous codes for the time being.

```
program guideAppDelphi;
```

```

{$APPTYPE CONSOLE}

{$R *.res}

uses
  System.SysUtils, ComObj, Windows,
  KikusuiPwxLib_TLB in 'KikusuiPwxLib_TLB.pas';

var
  hr: HRESULT;
  instr: IKikusuiPwx;
begin
  try
    { TODO -oUser -cConsole Main : Insert code here }
    instr:= CoKikusuiPwx.Create;
    hr:= instr.Initialize(
      'TCPIP::192.168.1.3::INSTR', true, true, 'QueryInstrStatus=1');

    hr:= instr.Close();

  except
    on E: Exception do
      WriteLn(E.ClassName, ': ', E.Message);
    end;
  end.

```

In this example codes, content of the main program is executed linearly. If the instrument is actually connected and the **Initialize** method call has succeeded, the program will finish silently. However, if a communication problem has occurred or the VISA library is not configured properly, an error code will be returned to HRESULT.

How to handle errors (exception) is explained later.

2-6 Repeated Capabilities, Output Collection

In case of IVI drivers for such as power supply or oscilloscope, the driver is designed assuming the instrument has multiple channels. Therefore for properties and methods that access instrument settings, there are a lot of cases that Repeated Capabilities (or Collection) are implemented. As for instrument drivers of DC power supplies, it is the **Output** collection.

For the case of KikusuiPwx IVI-COM driver, its concept is in **KikusuiPwxOutputs** and **KikusuiPwxOutput**. The plural name is the collection and singular name is each item (1 or more items) which may exist in the collection. In general an IVI instrument driver for DC power supply is designed assuming the instrument is a multi-track model.

They have the same name except for differences plural and singular forms. Like this, a component that has a plural name is generally called as Repeated Capabilities in the IVI spec. (Also called as Collection in COM terminology). The COM interface having plural name such as **IKikusuiPwxOutputs** normally has **Count**, **Name**, and **Item** properties (all read-only). **Count** property returns number of objects, **Name** property returns the name of the indexed object, and **Item** property returns reference to the object specified by a name. .

The following code example controls the output channel that is identified by "**Output0**" for the Kikusui Pwx series DC supply.

```

var
  ...
  { 変数を追加 }
  outputs: IKikusuiPwxOutputs;
  output: IKikusuiPwxOutput;
  ...

```

```

begin
  ...
  hr:= instr.get_Outputs( outputs);

  hr:= outputs.get_Item( 'output0', output);

  hr:= output.set_VoltageLevel( 20.0);
  hr:= output.set_CurrentLimit( 2.0);
  hr:= output.set_Enabled( true);
end;

```

Once the **IKikusuiPwxOutput** interface has been acquired, there is no difficulty at all. The **VoltageLevel** and the **CurrentLimit** properties set voltage level and current limit settings respectively. The **Enabled** property switches output ON/OFF state.

Notes:

- In contrast with Visual Basic or C#, when using COM interfaces derived directly from IUnknown (or incompatible with Automation) in Delphi, you cannot treat properties as if object's data members. To access properties, you need use accessor methods that have set_(when setting) or get_(when getting) prefixes. This is because all the methods and accessor methods commonly have the unified return value - HRESULT.
- Similarly, any methods having an imaginary return value do not allow to receive the value with assignment to the left side value (using a := operator), and the value must be received through the reference at the last parameter of each method call.

Mind the grammar for acquiring the **IKikusuiPwxOutput** interface. This example here acquires the **IKikusuiPwxOutputs** interface though the **Output** property of the **IKikusuiPwx** interface, then acquires **IKikusuiPwxOutput** interface by using the **Item** property.

Now mind the parameter passed to the **Item** property. This parameter specifies the name of the single Output object to be referenced. Actual available names (Output Name) are however different depending on drivers. For example, KikusuiPwx IVI-COM driver uses an expression like "**Output0**". However other drivers, even if being IviDCPwr class-compliant, may have different names. One instrument driver, for example, may use an expression like "**Channel1**". Although available names on a particular instrument driver are normally documented in the driver's online help, you can also check them out by writing some test codes shown below.

```

var
  ➔ add variables below
  n: longint;
  c: longint;
  name: WideString;

```

```

begin
  ...

  hr:= outputs.get_Count( &c);
  for n:=1 to c do
  begin
    outputs.get_Name( n, name);
    writeln (name);
  end;

  ...

```

The **Count** property returns number of single objects that the repeated capabilities have. The **Name** property returns the name of single object for the given index. The name is exactly the one that can be passed to the **Item** property as a parameter. In the above example, the codes iterate from the index 1 to **Count** by using the for block. Mind that the index numbers for the **Name** parameter is one-based, not zero-based.

3- Error Handling

In the previous examples, there was no error handling processed. However, setting an out-of-range value to a property or invoking an unsupported function may generate an error from the instrument driver. Furthermore, no matter how the application is designed and implemented robustly, it is impossible to avoid instrument I/O communication errors.

When using IVI-COM instrument drivers, every error generated in the instrument driver is transmitted to the client program as a COM exception. In case of Delphi, however, HRESULT has to be validated for each method call to check if a COM exception is generated.. Now let's change the example of setting voltage and current as follows.

```
try
  { TODO -oUser -cConsole Main : Insert code here }
  instr:= CoKikusuiPwx.Create;
  hr:= instr.Initialize(
    'TCPIP::192.168.1.5::INSTR',
    true,
    true,
    'QueryInstrStatus=1');
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= instr.get_Outputs( outputs);
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= outputs.get_Item( 'Output0', output);
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= output.set_VoltageLevel( 20.0);
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= output.set_CurrentLimit( 2.0);
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= output.set_Enabled( true);
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);

  hr:= instr.Close();
  if FAILED(hr) then
    raise EOLESysError.Create('', hr, 0);
except
  on E: EOLESysError do
    writeln( Format( 'Error:0x%08x', [E.ErrorCode]));
  on E: Exception do
    writeln(E.ClassName, ': ', E.Message);
end;
```


In this example, HRESULT value is validated after each method call. For example, if the name passed to the **Item** property is wrong, if an out-of-range value is passed to **VoltageLevel**, or if an instrument communication error is generated, a negative value will be returned as HRESULT. In the example, when an exception is generated, it create an **EOleSysError** object and raises it with **raise** statement. At the **except** block, it extracts the **ErrorCode** (the value of HRESULT) embedded in the thrown **EOleSysError** object, then create a simple message displaying the error code in the console.

Notes:

- The Delphi document says that a COM exception (HRESULT < 0) generated through COM interfaces throws an **EOleException** exception, however it is true only when using automation-compatible interfaces derived from IDispatch. Every COM interface defined by IVI-COM and VISA COM is a custom interface directly derived from IUnknown, therefore **EOleException** will not be thrown by framework.

4- Example Using Class Interface

Now we explain how to use class interfaces. By using class interfaces, you can swap the instruments without recompiling/relinking your application codes. In this case, however, IVI-COM instrument drivers for both pre-swap and post-swap models must be provided, and these drivers both must belong to the same instrument class. There is no interchangeability available between different instrument classes.

4-1 Virtual Instrument

What you have to do before creating an application that utilizes interchangeability features is create a virtual instrument. To realise interchangeability features, you should not write codes that are very specific to a particular IVI-COM instrument driver (e.g. creating an object instance directly as KikusuiPwx type) and should not write a specific VISA resource name such as "TCPIP::192.168.1.5::INSTR". Writing them directly in the application spoils interchangeability.

Instead, the IVI-COM specifications define methods to realise interchangeability by placing an external IVI configuration store. The application indirectly selects an instrument driver according to contents of the IVI Configuration Store, and accesses the indirectly loaded driver through the class interfaces.

The IVI Configuration Store is normally **C:/ProgramData/IVI Foundation/IVI /IviConfigurationStore.XML** file and is accessed through the IVI Configuration Server DLL. This DLL is mainly used by IVI instrument drivers and some VISA/IVI configuration tools, not by end-user applications. Instead, you can edit IVI driver configuration by using NI-MAX (NI Measurement and Automation Explorer) bundled with NI-VISA or IVI Configuration Utility bundled with KI-VISA.

Notes:

- As for how to edit virtual instrument settings using NI-MAX, refer to "IVI Instrument Driver Programming Guide (LabVIEW Edition or LabWindows/CVI Edition)".

This guidebook assumes that a virtual instrument having the logical name **mySupply** is already created, using KikusuiPwx driver, and using a VISA resource "TCPIP::192.168.1.5::INSTR".

4-2 Creating Application Project


Similarly to the example using specific interface, choose **File | New | Other...** menu and create a new **Console Application** project..

The project has no file names immediately after it is created. Choose **File | Save Project As...** menu then give **guideAppDelphi_DCPwr** as the project file name.

4-3 Importing Type Libraries

What you should do first after creating a new project is, import type libraries for the IVI-COM instrument driver that you want to use. Again use TLIBIMP.EXE.

Launch Command Prompt, and then use **cd** (change directory) command to go to the directory where the project files previously created are located. If the current directory display (using **dir** command) shows the project previously saved, it is the correct location.



```

C:\> Command Prompt
E:\_JunkWork\guide\delphi\2>dir
Volume in drive E is kamakura
Volume Serial Number is AA00-0550

Directory of E:\_JunkWork\guide\delphi\2

2012/06/13  17:25    <DIR>          .
2012/06/13  17:25    <DIR>          ..
2012/06/13  17:25                260 guideAppDelphi_DCPwr.dpr
2012/06/13  17:25                7,049 guideAppDelphi_DCPwr.dproj
2012/06/13  17:25                291 guideAppDelphi_DCPwr.dproj.local
                3 File(s)              7,600 bytes
                2 Dir(s)  82,988,748,800 bytes free


E:\_JunkWork\guide\delphi\2>

```

Figure 4-1 Working At Project Directory

Now execute the following command on the Command Prompt.

```
tlbimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviDCPwrTypeLib.dll"
```



```

C:\> Command Prompt
E:\_JunkWork\guide\delphi\2>tlbimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviDCPwrTypeLib.dll"
Embarcadero TLIBIMP Version 12.16581
Copyright(c) 1995-2010 Embarcadero Technologies, Inc.

Opening C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviDCPwrTypeLib.dll
Type library loaded ...
Created E:\_JunkWork\guide\delphi\2\IviDriverLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\2\IviDriverLib_TLB.pas
Created E:\_JunkWork\guide\delphi\2\IviDCPwrLib_TLB.dcr
Created E:\_JunkWork\guide\delphi\2\IviDCPwrLib_TLB.pas


E:\_JunkWork\guide\delphi\2>

```

Figure 4-2 Generating IviDCPwr TypeLib wrapper

Continuously execute the following command on the Command Prompt.

```
tlbimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviSessionFactory.dll"
```



```

C:\> Command Prompt

E:\_JunkWork\guide\delphi\2>tlibimp -P "C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviSessionFactory.dll"
Embarcadero TLIBIMP Version 12.16581
Copyright(c) 1995-2010 Embarcadero Technologies, Inc.

Opening C:/Program Files (x86)/IVI Foundation/IVI/Bin/IviSessionFactory.dll
Type library loaded ....
Created! E:\_JunkWork\guide\delphi\2\IviDriverLib_TLB.dcr
Created! E:\_JunkWork\guide\delphi\2\IviDriverLib_TLB.pas
Created! E:\_JunkWork\guide\delphi\2\IVISESSIONFACTORYLib_TLB.dcr
Created! E:\_JunkWork\guide\delphi\2\IVISESSIONFACTORYLib_TLB.pas

E:\_JunkWork\guide\delphi\2>_

```

Figure 4-3 Generating IviSessionFactory TypeLib wrapper

Again back to the Delphi IDE, add the files to your project. Right-click on `guideAppDelphi_DCPwr.exe` name in the Project Manager window to show the context menu, select **Add...** menu, select **IviDCPwrLib_TLB.pas** and **IVISESSIONFACTORYLib_TLB.pas** from the previously generated files, then add them to the project. It will be okay if the files have been added in the Project Manager




Figure 4-4 Adding TLB Wrappers

Open the main program in the project, add **ComObj** and **windows** units to the existing **uses** clause.

```

uses
  System.SysUtils, ComObj, windows,
  IviDCPwrLib_TLB in 'IviDCPwrLib_TLB.pas',
  IVISESSIONFACTORYLib_TLB in 'IVISESSIONFACTORYLib_TLB.pas';

```

4-4 Creating Object and Initializing Session

After completing type library import, write the code fragments in the button handler. (Here, write the complete codes including exception handling previously mentioned.)

```

uses
  System.SysUtils, ComObj, windows,

```

```

IviDCPwrLib_TLB in 'IviDCPwrLib_TLB.pas',
IVISessionFactoryLib_TLB in 'IVISessionFactoryLib_TLB.pas';

var
  hr : HRESULT;
  unk : IInterface;
  sf : IIVISessionFactory;
  instr : IIVIInstrument;
  outputs : IIVIInstrumentOutputs;
  output : IIVIInstrumentOutput;
begin
  try
    { TODO -oUser -cConsole Main : Insert code here }
    sf:= CoIVISessionFactory.Create;
    hr:= sf.CreateDriver( 'mySupply', unk);
    hr:= unk.QueryInterface(IID_IIVIInstrument, instr);

    hr:= instr.Initialize('mySupply', true, true, '');

    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= instr.get_Outputs( outputs);
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= outputs.get_Item( 'Track_A', output);
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= output.set_VoltageLevel( 20.0);
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= output.set_CurrentLimit( 2.0);
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= output.set_Enabled( true);
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

    hr:= instr.Close();
    if FAILED(hr) then
      raise EOLESysError.Create('', hr, 0);

  except
    on E: EOLESysError do
      writeln( Format( 'Error:0x%08x', [E.ErrorCode]));
    on E: Exception do
      writeln(E.ClassName, ': ', E.Message);
  end;
end.

```

Let's explain from the beginning.

4-5 Creating Object and Initializing Session

In contrast using specific interfaces, any dependency to specific components such as **KikusuiPwx** cannot be written. Instead, it creates an instance of **SessionFactory** object,

and indirectly create a driver object that is configured in the IVI Configuration Store by using **CreateDriver** method.

At first, notice that any type names beginning with Kikusui are not used. This example code no longer has dependency on KikusuiPwx.

Now create an **IviSessionFactory** object, then obtain the reference to **IiViSessionFactory** interface.

```
sf:= CoIviSessionFactory.Create;
```

Next, invoke the **CreateDriver** method passing the IVI Logical Name (Virtual Instrument). The created object is returned as a pointer to **IUnknown** interface, so prepare a variable **unk** as **IUnknown** receiving the pointer value.

```
hr:= sf.CreateDriver( 'mySupply', unk);
```

If succeeded to create a driver object then the returned HRESULT shall be 0. At the point of time **spUnk** is pointing to the **IUnknown** interface living inside the KikusuiPwx driver object. However we instead reference the **IiViDCPwr** interface storing into **instr**.

```
hr:= unk.QueryInterface(IID_IiViDCPwr, instr);
{ instr:= unk as IiViDCPwr; you can write like this too }
```

Notes:

- Because **CreateDriver** method has a capability to return arbitrary instrument driver and arbitrary instrument class interface, so it syntactically returns **IUnknown** pointer. In many instrument driver implementations, interfaces such as **IUnknown**, **IiViDriver**, **IiViDCPwr** are all equivalent pointer value, but as a COM's rule it is not always true. (An instrument driver implementation using aggregate may actually return different pointers.) Therefore in the above example, the **IUnknown** interface shall be referenced once, and explicitly reference the **IiViDCPwr** interface with **QueryInterface** method.
- In Delphi, only the **IUnknown** interface wrapper is renamed to **IInterface**.

If IVI Configuration Store is properly configured, the code will execute without generating exceptions. However, at this point of time, it has not communicate with the instrument yet. The DLL of IVI -COM driver is just loaded.

Then invoke **Initialize** method. At this point of time, communications with the instrument begins. The 1st parameter to **Initialize** method was originally a VISA address (VISA IO resource) but, here it shall be the IVI Logical Name. The IVI Configuration Store already knows the linked info concerning to this Logical Name, such as Hardware Asset, therefore the VISA address specified there will be actually applied.

```
hr:= instr.Initialize('mySupply', true, true, '');
```

As for **IiViDCPwr** class, the **Output** object of DC power supply is found in the **Outputs** collection. Similarly to the example of using specific interface, it obtains the reference to the single **Output** object from the collection. In this case, the interface type is **IiViDCPwrOutput** instead of **IKikusuiPwxOutput**.

```
hr:= instr.get_Outputs( outputs);
hr:= outputs.get_Item( 'Track_A', output);

hr:= output.set_voltageLevel( 20.0);
```

```
hr:= output.set_CurrentLimit( 2.0);  
hr:= output.set_Enabled( true);
```

Mind the parameter that is passed to **Item** parameter. This parameter specifies the name of single **Output** object that you want to reference to. In the example using specific interfaces it passed Physical Name that may be different by driver implementation basis, but not here. This example cannot use such Physical Names very specific to an instrument driver implementation (in fact it is possible to use but shall not to avoid spoiling interchangeability), so we use a Virtual Name.

The virtual name "**Track_A**" that is used in the above example is what specified to map to the physical name "**Output0**" in the IVI Configuration Store.

4-6 Exchanging Instrument

Example shown so far were set to use kipwx instrument driver as the virtual instrument configuration. Now what happens if changing the instrument to the one that is hosted by AgN57xx driver (Agilent N5700 series DC Power Supply)? In this case, you don't have to recompile/relink your application, however you have to change the configuration for IVI Logical Name (virtual instrument). Basically the configuration shall change:

- Software Module in Driver Session tab (kipwx→AgN57xx)
- map target of Virtual Names (Output0→Output1)
- IO Resource Descriptor in Hardware Asset (changing to the VISAaddress of post-swap instrument)
- Once the configuration is properly set, the above example will function with the post-swap instrument without having to recompile.

Once the configuration is properly set, the above example will function with the post-swap instrument without having to recompile.

Notes:

- For how to configure virtual instruments, refer to "IVI Instrument Driver Programming Guide (LabVIEW Edition or LabWindows/CVI Edition)".
- The interchangeability feature utilizing IVI class drivers does not guarantee the correct operation between pre-swapping and post-swapping instruments. Please make sure to confirm that your system correctly functions after swapping the instruments.

IVI Instrument Driver Programming Guide

Product names and company names that appear in this guidebook are trademarks or registered trademarks of their respective companies.

©2012 Kikusui Electronics Corp. All Rights Reserved.